
Chapter 11

Analysis of Variance
(One-Way)

We now develop a statistical procedure for comparing the means of two or
more groups, known as analysis of variance or ANOVA. These groups might
be the result of an experiment in which organisms are exposed to different
treatments. Alternately, the groups might be different species or different
age classes of the same species, populations in different locations, or different
genetic families. The test works by comparing the variance among the group
means to the variance of the observations within each group – if the variance
among group means is large (implying differences in their means) relative to
the variance within groups, the test is significant. This chapter will examine
tests for one-way ANOVA, in which a single factor like a treatment affects
the observations. More complex designs are possible in which several factors
may influence the observations and may also interact (see Chapter 14 and
19).

What do the data look like for a one-way ANOVA design? Suppose we
are interested in trapping bark beetles (Coleoptera: Curculionidae: Scolyti-
nae) using different chemical baits, which could involve the beetle’s sex
pheromones or odors of the trees they colonize. Suppose there are three
different baits (A, B, and C), with a = 3 denoting the number of treatments.
The baits are deployed on traps in the forest, with n = 5 replicate traps for
each bait type. A typical experimental design would establish 15 traps in the
forest, and then randomly assign a bait to each trap. After a period of time,
the traps would be checked and the number of insects caught in each trap
recorded (Table 11.1). Because the data are counts, it would not be normally

distributed but more likely have a Poisson or negative binomial distribution
(see Chapter 5). However, it is often possible to transform count data to
have a distribution closer to the normal by taking the square root or log of
the counts (see Chapter 15). The third column in Table 11.1 shows the count
data after applying a log transformation. The notation Yij is often used to
refer to the observations in ANOVA designs. The i subscript refers to the
group or treatment, while j is the observation within the treatment. For
example, Y13 refers to the third observation in the first treatment, which is
2.41.

Another one-way ANOVA design for bark beetles might simply look at
variability in their density across sites. Suppose there is a large collection
of study sites, and we randomly select five sites for trapping. Five traps are
deployed at each of the five sites and the number of beetles caught per trap
is recorded. Data for a study of this type are listed below, also with a log
transformation (Table 11.2). There appears to be substantial variability in
beetle abundance across sites, with Site 4 having very high beetle catches,
while Site 5 has low ones.

The data sets presented in this section represent balanced designs, be-
cause there are the same number of replicates for each treatment or group.
An unbalanced design would have an unequal number of replicates, pos-
sibly very unequal. We will present tests and theory for balanced designs
in this chapter, because this greatly simplifies the formulas and equations.
However, these results can be readily extended to unbalanced designs, and
unbalanced designs require no changes in the SAS programs presented.

Table 11.1: Example 1 - Bark beetles captured in a trapping experiment
comparing the attraction to different baits. There were three baits (A, B,
and C) and five replicate traps per bait treatment. Also shown are the log-
transformed counts (Yij) and subscript values (i, j), and some preliminary
one-way ANOVA calculations.

Treatment Count Yij = i j Ȳi· (Yij − Ȳi·)2
∑

(Yij − Ȳi·)2

log10(Count)
A 373 2.57 1 1 0.0441
A 126 2.10 1 2 0.0676
A 255 2.41 1 3 2.3600 0.0025 0.2110
A 138 2.14 1 4 0.0484
A 379 2.58 1 5 0.0484
B 25 1.40 2 1 0.0999
B 64 1.81 2 2 0.0088
B 62 1.79 2 3 1.7160 0.0055 0.1325
B 71 1.85 2 4 0.0180
B 54 1.73 2 5 0.0002
C 449 2.65 3 1 0.1832
C 249 2.40 3 2 0.0317
C 69 1.84 3 3 2.2220 0.1459 0.4581
C 199 2.30 3 4 0.0061
C 84 1.92 3 5 0.0912

Table 11.2: Example 2 - Bark beetles captured in a trapping study comparing
their abundance at five randomly chosen study sites. There were five replicate
traps per site. Also shown are the log-transformed counts (Yij) and subscript
values (i, j), and some preliminary one-way ANOVA calculations.

Site Count Yij = i j Ȳi· (Yij − Ȳi·)2
∑

(Yij − Ȳi·)2

log10(Count)
1 137 2.14 1 1 0.0164
1 101 2.00 1 2 0.0001
1 113 2.05 1 3 2.0120 0.0014 0.1598
1 48 1.68 1 4 0.1102
1 155 2.19 1 5 0.0317
2 156 2.19 2 1 0.0784
2 165 2.22 2 2 0.0625
2 652 2.81 2 3 2.4700 0.1156 0.4730
2 179 2.25 2 4 0.0484
2 757 2.88 2 5 0.1681
3 278 2.44 3 1 0.0376
3 197 2.29 3 2 0.0019
3 95 1.98 3 3 2.2460 0.0708 0.3419
3 395 2.60 3 4 0.1253
3 83 1.92 3 5 0.1063
4 2540 3.40 4 1 0.4956
4 613 2.79 4 2 0.0088
4 200 2.30 4 3 2.6960 0.1568 0.7600
4 251 2.40 4 4 0.0876
4 390 2.59 4 5 0.0112
5 18 1.26 5 1 0.0044
5 16 1.20 5 2 0.0000
5 11 1.04 5 3 1.1940 0.0237 0.0459
5 21 1.32 5 4 0.0159
5 14 1.15 5 5 0.0019

11.1 ANOVA models

We now examine the statistical models that are used in one-way ANOVA.
There are two models for one-way ANOVA, known as fixed or random effects
models, but sometimes called Model I and II. This classification is based on
how the groups in the design are defined or generated. We begin by defining
fixed and random effects, then present the statistical models and hypotheses
for each type.

11.1.1 Fixed and random effects

For groups generated by different treatments in an experiment, or purposely
chosen groups of organisms such as different species, sexes, or ages, the groups
are classified as fixed effects. They are called fixed effects because these
groups are the only ones of interest to the investigator, and the only ones
on which a statistical inference can be made (Littell et al. 1996, McCulloch
and Searle 2001). They are also incorporated in statistical models as fixed
parameters. Groups that are generated by a process of random sampling are
classified as a random effects (Littell et al. 1996, McCulloch and Searle
2001). For example, suppose we want to examine the fish populations in a
large number of lakes, and are interested in how body length varies across
lakes. If we randomly sample the lakes to be examined, from a large col-
lection of lakes, then lake would be classified as a random effect. In many
genetic experiments, families are chosen at random from a larger collection of
families, making family a random effect. Random effects are incorporated in
statistical models as random variables, typically with a normal distribution.

These definitions suggest a simple test for fixed vs. random
effects – if the groups are a random sample from a large collection
you have a random effect, otherwise a fixed effect. Although it is
usually possible to declare an effect as either fixed or random, in practice
it is sometimes difficult to decide. For example, suppose that a particular
organism occurs at only a small number of locations. If we randomly select
a subset of these locations to sample, seemingly implying a random effect,
the overall number of locations is still finite. In this scenario, location may
be better classified as a fixed effect.

11.1.2 Fixed effects model

Suppose that we want to model the observations in the bark beetle trap-
ping experiment, Example 1, where different baits are used. Recall that the
symbol Yij stand for the jth observation in the ith treatment group, where
i = 1, 2, 3 and j = 1, 2, 3, 4, 5. For example, Y11 = 2.57 and Y12 = 2.10, while
Y32 = 2.40 (see Table 11.1). One commonly used model for such a design is

Yij = µ+ αi + εij (11.1)

where µ is a parameter setting the grand mean (the overall mean) of the
observations, αi is the deviation from the grand mean caused by the ith
treatment (McCulloch and Searle 2001), and εij ∼ N(0, σ2). It is usually
assumed that

∑
αi = 0, i.e., the treatment effect terms sum to zero. The εij

term represents random departures from the mean value for the ith treat-
ment, due to natural variability among the observations. The εij values are
also assumed to be independent (Chapter 4). In practice, these parameters
would be unknown but could be estimated from the data. The same model
can be used to describe the observations for experiments with any number
of treatments (any a value) as well as replicates per treatments (any n),
as well as experiments where the number of observations is unequal among
treatments.

It follows that for the ith treatment, E[Yij] = µ + αi and V ar[Yij] = σ2,
using the rules for expected values and variances. Thus, for the ith treatment
we have Yij ∼ N(µ+ αi, σ

2). We can illustrate how the different parameters
work in this model with a diagram that plots the distribution for each group.
Suppose that we want to model an experiment similar to the bark beetle
trapping one, with a = 3 treatments. Suppose that µ = 2.1, α1 = 0.25,
α2 = −0.40, and α3 = 0.15, with σ2 = 0.1. Fig. 11.1 shows the distribution of
the observations in each treatment group. Note that the means for treatment
1 and 3 are shifted upward from the grand mean by their positive values of
αi, while the mean for treatment 2 is shifted downward by its negative value.
The distribution for each treatment has the same variance, namely σ2 = 0.1.

The usual objective in ANOVA is to test whether the means of the dif-
ferent groups are significantly different, implying there is treatment or group
effect. In terms of the fixed effects model, this amounts to testing whether
the αi values are significantly different from zero, because it is these param-
eters that produce shifts in the group means from the grand mean. More
formally, we are interested in testing the null hypothesis H0 : all αi = 0.

Under H0, all the groups have the same mean µ because the αi terms are
zero (Fig. 11.2). The alternative hypothesis would be H1 : some αi 6= 0, i.e.,
there is some treatment effect on some (perhaps all) groups (Fig. 11.1). We
will discuss how this null hypothesis is actually tested later in the chapter.

Figure 11.1: Fixed effects model for one-way ANOVA, under H1 : some
αi 6= 0.

Figure 11.2: Fixed effects model for one-way ANOVA, under H0 : all αi = 0.

11.1.3 Random effects model

Suppose that we now want to model the variability in bark beetle abundance
across different sites, such as in the Example 2 study. Let Yij stand for the jth
observation at the ith sampled site, with i = 1, 2, 3, 4, 5 and j = 1, 2, 3, 4, 5.
We have Y11 = 4.92, Y12 = 4.62, and so forth (see Table 11.2). A common
statistical model for this design is

Yij = µ+ Ai + εij (11.2)

where µ is again a parameter setting the grand mean of the observations, with
Ai a random deviation from the grand mean due to the ith site (McCulloch
and Searle 2001), and εij ∼ N(0, σ2). It is assumed that Ai is normally
distributed with mean zero and variance σ2

A, or Ai ∼ N(0, σ2
A). Note that

in the random effects model the group effect is indeed a random variable,
one whose variance is unknown but can be estimated from the data. The
variances σ2

A and σ2 are collectively called the variance components of the
model.

For the ith group sampled, it can be shown that E[Yij] = µ + Ai and
V ar[Yij] = σ2, using the rules for expected values and variances. Thus, for
the ith treatment we have Yij ∼ N(µ + Ai, σ

2). Because the Ai values are
themselves random quantities, however, the expected value is itself a random
quantity and would differ for each group sampled. We again illustrate how
the model works using a diagram showing the distribution for each group.
Suppose that we want to model a study similar to the second bark beetle one
(Table 11.2), with a = 5 sites randomly selected from a larger collection of
sites. Suppose that µ = 2.1 and σ2 = 0.1. The first time we did this study,
we might see a pattern like Fig. 11.3. If we redid the study and randomly
selected another five sites, we would get a different pattern (Fig. 11.4). This
illustrates that this model is not static like the fixed effects one, but instead
would vary with the sites actually sampled. In the random effects model,
we are usually interested in testing whether the variance of Ai is zero vs.
greater than zero, or H0 : σ2

A = 0 vs. H1 : σ2
A > 0. Under H0 : σ2

A = 0, all
the Ai values must be zero (to give σ2

A = 0), and so all the groups have the
same mean µ. A plot of the model under H0 would therefore be similar to
Fig. 11.2. This null hypothesis is tested in the same way as the one for the
fixed effects model (see below).

Figure 11.3: Random effects model for one-way ANOVA, for the first time
sites are sampled.

Figure 11.4: Random effects model for one-way ANOVA, for the second time
sites are sampled.

11.2 Hypothesis testing for ANOVA

We now develop a statistical test for the null hypotheses in both fixed and
random effects models, either H0 : all αi = 0 or H0 : σ2

A = 0. We will first
present the test and explain how it works in terms of different estimates of
the variance, then later show it is another example of a likelihood ratio test.

11.2.1 Sums of squares and mean squares

Suppose the data are described by a fixed effects model, for which the hy-
potheses are H0 : all αi = 0 vs. H1 : some αi 6= 0. It is clear that if H1 is
true, then the observations for the different groups will be shifted from the
grand mean, as shown in Fig. 11.1, and in particular Yij ∼ N(µ + αi, σ

2)
for each group. For a random effects model, we have H0 : σ2

A = 0 vs.
H1 : σ2

A > 0. If H1 is true, we would also expected the observations for the
different groups to be shifted away from the grand mean (Fig. 11.3), and
in particular Yij ∼ N(µ + Ai, σ

2). How can we estimate this shift in actual
data? How large must this shift be to be judged statistically significant?

We begin by calculating the means for each group using the data. These
are labeled as Ȳi· and are called group means. The ‘·’ subscript implies the
mean was calculated using all the observations in that group (j = 1, 2, . . . , n).
We then calculate the mean of the group means, called the grand mean and
labeled as ¯̄Y . If the ith group is shifted from the grand mean, we can measure
this shift using the quantity Ȳi· − ¯̄Y . In fact, this quantity estimates αi for
the ith group, and so is a direct measure of any group effect (see Section 11.3
on maximum likelihood estimation). If these quantities are small then this
suggests H0 might be true, whereas if they are large this provides evidence for
H1. We can obtain a single measure of these shifts by squaring and summing
them across all groups, to obtain a quantity called the sum of squares among
groups or SSamong, because it measures variation in the observations among
groups:

SSamong = n

a∑
i=1

(Ȳi· − ¯̄Y )2. (11.3)

Note the sample size n in this expression, which we will justify below. To
make this quantity more concrete, we will calculate SSamong for Example 1,
the bark beetle trapping experiment. We first calculate the sample mean for

each group for the log-transformed data, as shown in Table 11.1. Then, the
grand mean is estimated using the mean of these means, or

¯̄Y =

∑a
i=1 Ȳi·
a

=
2.3600 + 1.7160 + 2.2220

3
=

6.2980

3
= 2.0993. (11.4)

We then have

SSamong = n

a∑
j=1

(Ȳi· − ¯̄Y )2 (11.5)

= 5
[
(2.3600− 2.0993)2 + (1.7160− 2.0993)2 + (2.2220− 2.0993)2

]
(11.6)

= 5 [0.0680 + 0.1469 + 0.0151] (11.7)

= 1.1500 (11.8)

SSamong has a−1 degrees of freedom, where a is the number of groups. There

are a − 1 degrees of freedom because there are a terms of the form Ȳi· − ¯̄Y
in the sum of squares, but these sum to zero so there are really only a − 1
independent terms (similar to the n − 1 degrees of freedom for the sample
variance s2). The next step is to convert SSamong to a sample variance,
dividing it by a− 1. This quantity is called the mean square among groups:

MSamong =
SSamong
a− 1

=
n
∑a

j=1(Ȳi· − ¯̄Y )2

a− 1
. (11.9)

For the bark beetle experiment, we find that

MSamong =
SSamong
a− 1

=
1.1500

3− 1
= 0.5750. (11.10)

So, what variance does MSamong estimate? If H0 is true and there are no
group effects, we would expect Ȳi· to have a variance of σ2/n, because it is
a sample mean composed of n observations in the ith group (which have a
variance of σ2). MSamong estimates this variance multiplied by n, because
of the n term in numerator, and so actually estimates nσ2/n = σ2. On the
other hand, if H1 is true then there are group effects, and we would expect the
group means to be shifted away from the grand mean. This should increase
the size of MSamong. Thus, MSamong estimates σ2 if H0 is true but
becomes larger if H1 is true.

We next develop an estimate of the variance σ2 that is free of any effects,
fixed or random. This variance estimate is based on a quantity called the
sum of squares within groups or SSwithin, because it measures variation of
the observations within each group. It is defined by the formula

SSwithin =
a∑
i=1

n∑
j=1

(Yij − Ȳi·)2 (11.11)

=
n∑
j=1

(Y1j − Ȳ1·)
2 + . . .+

n∑
j=1

(Yaj − Ȳa·)2. (11.12)

It has a(n− 1) degrees of freedom, because there are a sum of squares terms
each with n − 1 degrees of freedom. We can obtain an estimate of σ2 by
dividing this sum of squares by its degrees of freedom, to obtain the mean
square within groups:

MSwithin =
SSwithin
a(n− 1)

=

∑a
i=1

∑n
j=1(Yij − Ȳi·)2

a(n− 1)
. (11.13)

This quantity estimates σ2 because it simply averages estimates of σ2 for
each group. With some rearrangement, we can write MSwithin as

MSwithin =

∑a
i=1

∑n
j=1(Yij − Ȳi·)2

a(n− 1)
(11.14)

=

∑
(Y1j − Ȳ1·)

2 + . . .+
∑

(Yaj − Ȳa·)2

a(n− 1)
(11.15)

=

∑
(Y1j−Ȳ1·)2
n−1

+ . . .+
∑

(Yaj−Ȳa·)2
n−1

a
(11.16)

=
s2

1 + . . .+ s2
a

a
. (11.17)

Each term in the numerator of this expression is the sample variance s2

for each group, which is then averaged across all groups to yield an overall
or pooled estimate of σ2. The word ‘pooled’ in statistics often indicates a
combined estimate of a variance. It can also be shown that E[MSwithin] = σ2,
regardless of any group effects.

We now calculate MSwithin for the bark beetle experiment. We first need
to calculate the quantity (Yij − Ȳi·)2 for the observations in each group and

then sum these for each group (see Table 11.1). Summing these quantities
in turn across all groups, we obtain

SSwithin = 0.2110 + 0.1325 + 0.4581 = 0.8016. (11.18)

(11.19)

We then have

MSwithin =
SSwithin
a(n− 1)

=
0.8016

3(5− 1)
= 0.0668. (11.20)

(11.21)

There is one more sum of squares that can be calculated in one-way
ANOVA, the total sum of squares. It is defined as

SStotal =
a∑
i=1

n∑
j=1

(Yij − ¯̄Y )2. (11.22)

It measures the variability of the observations around the grand mean of the
data ( ¯̄Y ) and has an − 1 degrees of freedom. Applying this formula to the
Example 1 data set, we obtain SStotal = 1.9516 after much calculation.

An interesting feature of the sum of squares is that they add to the total
sum of squares, as do the degrees of freedom. In particular, we have

SSamong + SSwithin = SStotal (11.23)

and
(a− 1) + a(n− 1) = an− 1. (11.24)

Thus, the sum of squares and degrees of freedom can be neatly partitioned
into components corresponding to among group and within group variation.
We will illustrate this relationship further in the section below on ANOVA
tables.

11.2.2 F statistic and distribution

We next describe the statistic used to test H0 : all αi = 0 for the fixed effect
model, and H0 : σ2

A = 0 for the random effects one. It is simply the ratio of
MSamong and MSwithin, or

Fs =
MSamong
MSwithin

. (11.25)

If H0 is true for either model, both MSamong and MSwithin estimate σ2 and
we would expect their ratio, Fs, to be small and on the order of one. However,
if H0 is false and H1 is true, we would expect MSamong to become larger and
Fs to increase. We would therefore reject H0 for large values of Fs.

To complete our testing procedure and find P values, we need to know the
distribution of Fs under H0. It turns out this statistic has an F distribution
under H0, whose shape and location is governed by two parameters, the de-
grees of freedom for MSamong and MSwithin. These are called the numerator
and denominator degrees of freedom, which we abbreviate as df1 and df2. In
particular, for one-way ANOVA we have df1 = a − 1 and df2 = a(n − 1).
Figure 11.5 shows the F distribution for three different sets of parameter
values. Note that distribution can have a maximum at y = 0 for small values
of df1, while larger values of df2 decrease the probability in the right tail of
the distribution.

Figure 11.5: The F distribution for three different sets of parameter values

Table F gives the quantiles of the F distribution for different values of the
degrees of freedom and the cumulative probability p. Statistical tests that
make use of the F distribution are typically called F tests.

Calculating the test statistic Fs for the bark beetle experiment, we have

Fs =
MSamong
MSwithin

=
0.5750

0.0668
= 8.6078, (11.26)

with df1 = a− 1 = 3− 1 = 2 and df2 = a(n− 1) = 3(5− 1) = 12.
As with previous tests, we seek acceptance and rejection regions for a

particular value of α, the Type I error rate. In particular, we seek a quantity
cα,df1,df2 such that

P [0 < Fs < cα,df1,df2 ] = 1− α. (11.27)

The region is of this form because the test is designed to reject H0 for large
values of Fs, and accept it for small ones. To find cα,df1,df2 , we look in Table
F for the column corresponding to 1 − p = α, for the appropriate degrees
of freedom. The acceptance region would therefore be (0, cα,df1,df2), and we
would reject H0 if Fs lies outside this region.

For α = 0.05, df1 = 2, and df2 = 12, we see from Table F that c0.05,2,12 =
3.885. Our acceptance region is therefore (0, 3.885), and we reject H0 at the
α = 0.05 level if Fs ≥ 3.885 (Fig. 11.6). We see this is the case because
Fs = 8.6078 > 3.885. We can continue this process for increasingly smaller
α and eventually find that for α = 0.005 we can still reject H0, but not for
α = 0.001. We therefore have P < 0.005 for this test, because α = 0.005
is the smallest value of α for which we can reject H0 (see Chapter 10). An
F test in ANOVA would often be reported as follows: ‘There was a highly
significant difference among the different baits in the number of bark beetles
trapped (F2,12 = 8.6078, P < 0.005).’ Note that the degrees of freedom are
given as subscripts.

11.2.3 ANOVA tables

We can organize the different sum of squares and mean squares into an
ANOVA table. It lists the different sources of variation in the data (among,
within, and total), their degrees of freedom, sums of squares and mean
squares, and then the F statistic and its P value. Table 11.3 shows the
general layout of such a table for one-way ANOVA designs, while Table 11.4
gives the results for the Example 1 analysis. Note the additive relationship
for the degrees of freedom and sum of squares.

Figure 11.6: Acceptance and rejection regions for α = 0.05

Table 11.3: General ANOVA table for one-way designs with a groups and n observations per group, showing
formulas for different mean squares and the F test.

Source df Sum of squares Mean square Fs
Among a− 1 SSamong = n

∑a
i=1(Ȳi· − ¯̄Y )2 MSamong = SSamong/(a− 1) MSamong/MSwithin

Within a(n− 1) SSwithin =
∑a

i=1

∑n
j=1(Yij − Ȳi·)2 MSwithin = SSwithin/a(n− 1)

Total an− 1 SStotal =
∑a

i=1

∑n
j=1(Yij − ¯̄Y )2

Table 11.4: ANOVA table for the Example 1 data set, including a P value for the test.

Source df Sum of squares Mean square Fs P
Among 2 1.1500 0.5750 8.6078 < 0.005
Within 12 0.8016 0.0668
Total 14 1.9516

11.2.4 One-way ANOVA for Example 1 - SAS demo

The same calculations for the bark beetle experiment can be carried out in
SAS using proc glm (SAS Institute Inc. 2018). This procedure is primarily
intended for fixed effects ANOVA models, with proc mixed the best choice
for random effects models. However, the F test would be the same in either
procedure.

We will also use SAS and proc gplot (SAS Institute Inc. 2016) to visualize
the data. The basic idea is to plot, for each treatment group, the individual
data points along with their mean (Ȳ ) ± one standard error (s/

√
n). These

plots are useful for comparing the relative effects of the treatments, a con-
cept called effect size, as well as the variability of the observations. Effect
size is used to judge the biological significance of the treatments – are the
differences among the treatments biologically meaningful? This is distinct
from the statistical significance of the ANOVA. For example, you could ob-
serve large differences among the treatment means that could be biological
significant, but the F test could be non-significant because the data were
highly variable. Conversely, the differences among the means could be small
and not biologically meaningful, but the F test could be significant because
n is large, and so the test can detect even small differences.

The SAS program for one-way ANOVA is a bit more complicated than
previous programs, so we will examine it a section at a time. The first step
is to read in the observations using a data step, with one variable denoting
the treatment (treat) and a second the number of beetles captured (count).
As discussed earlier, it is common to log-transform count data, and so we
generate a variable y that is the log10 (log base 10) of count. The data step
is followed by a print statement to print the data set. See section below.

* bark_beetle_experiment.sas;

title "One-way ANOVA for bark beetle trapping experiment";

data bark_beetle;

input treat $ count;

* Apply transformations here;

y = log10(count);

datalines;

A 373

A 126

A 255

etc.

C 199

C 84

;

run;

* Print data set;

proc print data=bark_beetle;

run;

We next plot the data using proc gplot (SAS Institute Inc. 2016). The
plot statement tells gplot to plot the variable y on the y-axis and treat on
the x-axis of the plot. The appearance of the points is controlled by the
symbol1 statement, which among other things specifies that the points be
plotted along with their means ± one standard error, with the means joined
by a line, using the option i=std1mjt. Other options in the symbol statement
control the type and size of the points, and line width. The vaxis=axis1 and
haxis=axis1 options control the visual appearance of the x- and y-axes. See
below.

* Plot means, standard errors, and observations;

proc gplot data=bark_beetle;

plot y*treat=1 / vaxis=axis1 haxis=axis1;

symbol1 i=std1mjt v=star height=2 width=3;

axis1 label=(height=2) value=(height=2) width=3 major=(width=2) minor=none;

run;

The next section of the program conducts the one-way ANOVA and F test
using proc glm. The plots=diagnostics option generates graphs that are used
to examine some of the assumptions of ANOVA – we will defer their discus-
sion to Chapter 15. The class statement tells SAS that the variable treat is
the one that defines different groups in the ANOVA (see listing below). The
model statement basically tells SAS the form of the ANOVA model. Recall
that the model for fixed effects one-way ANOVA is given by the equation

Yij = µ+ αi + εij. (11.28)

If we equate Yij with y, and αi with treat, we see there are similarities between
the fixed effects model and the SAS model statement. In fact, SAS assumes
you want a grand mean µ unless otherwise specified, as well as the error term
εij. As we examine more complex ANOVA models in later chapters, we will
see there is nearly a one-to-one correspondence between these models and
the corresponding SAS model statement.

* One-way ANOVA with all fixed effects;

proc glm plots=diagnostics data=bark_beetle;

class treat;

model y = treat;

* Calculate means for each group;

means treat;

run;

The means statement causes glm to calculate means for each treat group.
The complete SAS program and output are listed below. The output

shows the same F test in three different locations within the proc glm output
(Fig. 11.9). The first is in a format resembling an ANOVA table, and then
two other times corresponding to Type I and III sums of squares. These
are different ways of calculating the sums of squares and tests, with Type
III sums of squares more generally useful for ANOVA designs. For one-
way ANOVA the results are the same, and we see that there was a highly
significant difference among groups (F2,12 = 8.60, P = 0.0048). Inspection of
the graph (Fig. 11.8) and means suggests that treatment A caught the most
beetles, followed by C and then B.

SAS Program

* bark_beetle_experiment.sas;

title "One-way ANOVA for bark beetle trapping experiment";

data bark_beetle;

input treat $ count;

* Apply transformations here;

y = log10(count);

datalines;

A 373

A 126

A 255

A 138

A 379

B 25

B 64

B 62

B 71

B 54

C 449

C 249

C 69

C 199

C 84

;

run;

* Print data set;

proc print data=bark_beetle;

run;

* Plot means, standard errors, and observations;

proc gplot data=bark_beetle;

plot y*treat=1 / vaxis=axis1 haxis=axis1;

symbol1 i=std1mjt v=star height=2 width=3;

axis1 label=(height=2) value=(height=2) width=3 major=(width=2) minor=none;

run;

* One-way ANOVA with all fixed effects;

proc glm plots=diagnostics data=bark_beetle;

class treat;

model y = treat;

* Calculate means for each group;

means treat;

run;

quit;

Figure 11.7: bark beetle experiment.sas - proc print

Figure 11.8: bark beetle experiment.sas - proc gplot

Figure 11.9: bark beetle experiment.sas - proc glm

11.2.5 One-way ANOVA for Example 2 - sample cal-
culation

We will conduct an F test for our second data set, involving a study of bark
beetles trapped at five different sites (a = 5) selected at random from a
collection of sites, with five traps per site (n = 5). This implies a random
effects model, and we are therefore interested in testing H0 : σ2

A = 0 vs.
H1 : σ2

A > 0. Some preliminary calculations for the F test are shown in
Table 11.2. We first find the mean Ȳi· for each site, then calculate the grand
mean as the average of the site means:

¯̄Y =

∑a
i=1 Ȳi·
a

(11.29)

=
2.0120 + 2.4700 + 2.2460 + 2.6960 + 1.1940

5
(11.30)

=
10.6180

5
= 2.1236. (11.31)

We then have

SSamong = n
a∑
j=1

(Ȳi· − ¯̄Y )2 (11.32)

= 5
[
(2.0120− 2.1236)2 + . . .+ (1.1940− 2.1236)2

]
(11.33)

= 5 [0.0125 + 0.1200 + 0.0150 + 0.3276 + 0.8642] (11.34)

= 6.6965 (11.35)

We next calculate MSamong:

MSamong =
SSamong
a− 1

=
6.6965

5− 1
= 1.6741. (11.36)

(11.37)

Now we find SSwithin, first calculating (Yij− Ȳi·)2 for the observations in each
group and then summing these for each group (see Table 11.2). Summing
these quantities in turn across all groups, we obtain

SSwithin = 0.1598 + 0.4730 + 0.3419 + 0.7600 + 0.0459 = 1.7806. (11.38)

(11.39)

We then have

MSwithin =
SSwithin
a(n− 1)

=
1.7806

5(5− 1)
= 0.0890. (11.40)

(11.41)

Calculating the test statistic Fs, we obtain

Fs =
MSamong
MSwithin

=
1.6741

0.0890
= 18.8101, (11.42)

(11.43)

with df1 = a − 1 = 4 − 1 = 4 and df2 = a(n − 1) = 5(5 − 1) = 20. From
Table F, we find P < 0.001. The variance among sites was highly significant
(F4,12 = 18.8101, P < 0.001.

11.2.6 One-way ANOVA for Example 2 - SAS demo

We can carry out the F test as well as estimate the variance components
(σ2

A and σ2) for the random effects model using SAS. The first section of the
program involving the data step and gplot graph is similar to the fixed effects
program. The next section of the program fits the random effects model to
the data and conducts the F test, using proc mixed (see listing below). As
before, the class statement tells SAS that the variable site is the one that
defines different groups in the ANOVA. Now recall that the model for random
effects one-way ANOVA is given by the equation

Yij = µ+ Ai + εij. (11.44)

Note that Ai corresponds to site in the bark beetle study. In proc mixed,
fixed effects in the model are placed in a model statement, while any random
effects are listed in a random statement (SAS Institute Inc. 2018). Because
our random effects model only has one random effect, site, this is listed in
the random statement. There are no fixed effects in this model, so the model

statement lists nothing after the equals sign. The option ddfm=kr specifies a
general method of calculating the degrees of freedom that works well under
many circumstances, including more complicated models.

* One-way ANOVA with random effects - F test;

proc mixed method=type3 data=bark_beetle;

class site;

model y = / ddfm=kr;

random site;

run;

* One-way ANOVA with random effects - variance components;

proc mixed cl plots=residualpanel data=bark_beetle;

class site;

model y = / ddfm=kr;

random site;

run;

Why is proc mixed invoked twice in this program? The first one generates
the F statistic for testing H0 : σ2

A = 0 vs. H1 : σ2
A > 0, using the option

method=type3. This is not the default in proc mixed, which appears more
designed to estimate the variance components in random effects (Littell et
al. 1996). If we drop this option, as in the second proc mixed statement, we

get only these estimates and no F test. Confidence intervals for the variance
components are requested using the cl option. The variance components
estimated in the second proc mixed using a version of maximum likelihood,
the preferred method of estimating these quantities.

The complete SAS program and output are listed below. The F test
found in the first proc mixed output (Fig. 11.12) was highly significant
(F4,12 = 18.77, P < 0.0001), suggesting σ2

A > 0. The second proc mixed

output provides estimates and confidence intervals for the two variance com-
ponents (Fig. 11.13). We have σ̂2

A = 0.3174 for which the 95% confi-
dence interval is (0.1093, 3.1458), and σ̂2 = 0.0893 with confidence interval
(0.0523, 0.1863). From these results, we see that the variance among sites
was greater than the variance within sites (0.3174 > 0.0893). This can also
be seen in Fig. 11.11 – beetle numbers vary considerably among sites relative
to within them.

SAS Program

* bark_beetle_random.sas;

title "One-way ANOVA for bark beetle sampling study";

data bark_beetle;

input site $ count;

* Apply transformations here;

y = log10(count);

datalines;

1 137

1 101

1 113

1 48

1 155

2 156

2 165

2 652

2 179

2 757

3 278

3 197

3 95

3 395

3 83

4 2540

4 613

4 200

4 251

4 390

5 18

5 16

5 11

5 21

5 14

;

run;

* Print data set;

proc print data=bark_beetle;

run;

* Plot means, standard errors, and observations;

proc gplot data=bark_beetle;

plot y*site=1 / vaxis=axis1 haxis=axis1;

symbol1 i=std1mjt v=star height=2 width=3;

axis1 label=(height=2) value=(height=2) width=3 major=(width=2) minor=none;

run;

* One-way ANOVA with random effects - F test;

proc mixed method=type3 data=bark_beetle;

class site;

model y = / ddfm=kr;

random site;

run;

* One-way ANOVA with random effects - variance components;

proc mixed cl plots=residualpanel data=bark_beetle;

class site;

model y = / ddfm=kr;

random site;

run;

quit;

etc.

Figure 11.10: bark beetle random.sas - proc print

Figure 11.11: bark beetle random.sas - proc gplot

Figure 11.12: bark beetle random.sas - proc mixed (1)

Figure 11.13: bark beetle random.sas - proc mixed (2)

11.3 Maximum likelihood estimates

This section sketches how the parameters in one-way ANOVA can be esti-
mated using maximum likelihood. Recall that the likelihood for a random
sample of three observations (Y1 = 4.5, Y2 = 5.4, Y2 = 5.3) from a normal
distribution (see Chapter 8) was of the form

L(µ, σ2) =
1√

2πσ2
e−

1
2

(4.5−µ)2

σ2 × 1√
2πσ2

e−
1
2

(5.4−µ)2

σ2 × 1√
2πσ2

e−
1
2

(5.3−µ)2

σ2 .

(11.45)

We found maximum likelihood estimates of the normal distribution param-
eters by maximizing this quantity with respect to µ and σ2.

Suppose now we have a data set that can be modeled using the fixed
effects one-way ANOVA model, in particular

Yij = µ+ αi + εij. (11.46)

This model has a number of parameters to estimate, such as µ, αi for i =
1, 2, . . . , a, and σ2. What would the likelihood function look like for these
data? Consider the first group for the bark beetle experiment (Example 1),
for which we have Y11 = 2.576, Y12 = 2.10, Y13 = 2.41, Y14 = 2.14, and
Y15 = 2.58. For the first group the model assumes that Y1j ∼ N(µ+ α1, σ

2),
and so the likelihood would be

L1 =
1√

2πσ2
e−

1
2

(2.57−(µ+α1))
2

σ2 × 1√
2πσ2

e−
1
2

(2.10−(µ+α1))
2

σ2 × 1√
2πσ2

e−
1
2

(2.41−(µ+α1))
2

σ2

(11.47)

× 1√
2πσ2

e−
1
2

(2.14−(µ+α1))
2

σ2 × 1√
2πσ2

e−
1
2

(2.58−(µ+α1))
2

σ2 .

(11.48)

The likelihood L2 for the second group would be similar, except that Y2j ∼
N(µ + α2, σ

2), and L3 similarly defined. The overall likelihood would then
be defined as

L(µ, α1, α2, α3, σ
2) = L1 × L2 × L3. (11.49)

Finding the maximum likelihood estimates involves maximizing this quan-
tity with respect to the parameters µ, α1, α2, α3, and σ2. The likelihood for

designs with any number of treatment groups and replicates would be simi-
lar. Using a bit of calculus to find the maximum, it can be shown that the
maximum likelihood estimates of these parameters, in general, are

µ̂ = ¯̄Y, (11.50)

α̂i = Ȳi· − ¯̄Y, (11.51)

and

σ̂2 =

∑n
i=1

∑n
j=1(Yij − Ȳi·)2

a(n− 1)
= MSwithin. (11.52)

(McCulloch & Searle 2001). These estimators seem quite reasonable. They
use the grand mean of the data, ¯̄Y , to estimate the grand mean µ of the
model, and the difference between the ith group mean and the grand mean,
Ȳi· − ¯̄Y , to estimate the deviation from the group mean αi. Note that σ̂2

is equal to MSwithin, which we have already encountered in our ANOVA
calculations.

Suppose now we have a data set suited to the random effects model, in
particular

Yij = µ+ Ai + εij. (11.53)

This model has three parameters to be estimated: µ, σ2
A, and σ2. The

likelihood for this model is more complex because of the random effect Ai, but
one can show that the maximum likelihood estimators of these parameters
are

µ̂ = ¯̄Y, (11.54)

σ̂2
A =

MSamong −MSwithin
n

, (11.55)

and
σ̂2 = MSwithin. (11.56)

An intuitive explanation of the formula for σ̂2
A is that MSamong incorporates

variance from both Ai and εij, while MSwithin only has εij. Subtracting
MSwithin from MSamong leaves only the variance due to Ai, so that the nu-
merator of this expression estimates nσ2

A. We then divide by n to obtain an
estimate of σ2

A.
Suppose that for an unusual data set we obtain MSamong < MSwithin,

implying a negative estimate of σ̂2
A = 0 according to the above equation. An

inherent feature of maximum likelihood is that is restricts variance compo-
nents to plausible values (McCulloch & Searle 2001), so in this case it would

simply say that σ̂2
A = 0, the smallest possible nonnegative value. This would

be reflected in the SAS output for proc mixed, which would report that the
variance component in question was zero. The estimates presented here are
actually obtained using a variant of maximum likelihood called restricted
maximum likelihood or REML. This method is the default in SAS, and has
some theoretical advantages over straight maximum likelihood (McCulloch
and Searle 2001).

11.4 F test as a likelihood ratio test

The F test in one-way ANOVA can be derived as a likelihood ratio test, simi-
lar to the development of the t test in Chapter 10. We first find the maximum
likelihood estimates of various parameters under H1 vs. H0, where the pa-
rameters under consideration are the ANOVA model parameters. Recall that
the observations in the fixed effects model are described as

Yij = µ+ αi + εij (11.57)

where µ is the grand mean, αi is the effect of the ith treatment, and εij ∼
N(0, σ2). This is the statistical model under the alternative hypothesis,
where αi 6= 0 for some i. Under H0 : all αi = 0, the model reduces to
just

Yij = µ+ εij. (11.58)

We would need to find the maximum likelihood estimates under both H1 (see
previous section) and H0, as well as LH0 and LH1 , the maximum height of
the likelihood function under H0 and H1. We would then use the likelihood
ratio test statistic

λ =
LH0

LH1

. (11.59)

It can be shown that there is a one-to-one correspondence between −2 ln(λ)
and Fs in one-way ANOVA, and so the F test is actually a likelihood ratio test
(McCulloch & Searle 2001). A similar argument can be made to justify the F
test for the random effects model. Like all likelihood ratio tests, large values
of the test statistic −2 ln(λ) or Fs indicate a lower value of the likelihood
under H0 relative to H1, and thus a poorer fit of the H0 model.

11.5 One-way ANOVA and two-sample t tests

There is an alternative to one-way ANOVA when there are only two groups
to be compared, the two-sample t test. Let µ1 be the mean of the first group
and µ2 the second one, and suppose that the two groups have the same
variance σ2 and sample size n. We are interested in testing H0 : µ1 = µ2 vs.
H1 : µ1 6= µ2, to determine if there are differences in the means of the two
groups. Under H0, the test statistic

Ts =
(Ȳ1· − Ȳ2·)√

s21+s22
2

∼ t2(n−1). (11.60)

Here Ȳ1· and Ȳ2· are the sample means for each group, and s2
1 and s2

2 the
sample variances. For a Type I error rate of α, the acceptance region of the
test would be the interval (−cα,2(n−1), cα,2(n−1)), where cα,2(n−1) is determined
using Table T (see Chapter 10). We would reject H0 if it falls on the edge or
outside this interval. There are also versions of this test statistic for unequal
sample sizes.

Although a two-sample t test is often used for comparing two groups, in
the form above it is equivalent to the F test in one-way ANOVA. To see
this, note that T 2

s = Fs for one-way ANOVA with two groups. It can also
be shown that the acceptance and rejection regions are the same for the
two tests. Unlike ANOVA, though, a two-sample t test can also be used for
one-tailed alternative hypotheses, such as H1 : µ1 > µ2 or H1 : µ1 < µ2.
The procedure is similar to one-sample t tests for one-tailed alternatives (see
Chapter 10).

11.5.1 Two-sample t test for Example 1 - SAS demo

We can illustrate this test by comparing treatment A and B in the Example 1
study, deleting the data for the third treatment. See SAS program and output
below. The data and proc gplot portions of the program are similar to our
previous one-way ANOVA code. The two-sample t test is carried out using
proc ttest (SAS Institute Inc. 2018), with the class statement indicating
the variable that codes for different groups (treat), while the var statement
designates the dependent variable (y). From Fig. 11.16, we see there was
a highly significant difference between treatment A and B (t8 = 4.90, P =
0.0012), with treatment A catching more beetles than B (Fig. 11.15).

SAS Program

* bark_beetle_experiment_ttest.sas;

title "Two-sample t test for bark beetle trapping experiment";

data bark_beetle;

input treat $ count;

* Apply transformations here;

y = log10(count);

datalines;

A 373

A 126

A 255

A 138

A 379

B 25

B 64

B 62

B 71

B 54

;

run;

* Print data set;

proc print data=bark_beetle;

run;

* Plot means, standard errors, and observations;

proc gplot data=bark_beetle;

plot y*treat=1 / vaxis=axis1 haxis=axis1;

symbol1 i=std1mjt v=star height=2 width=3;

axis1 label=(height=2) value=(height=2) width=3 major=(width=2) minor=none;

run;

* Two-sample t test;

proc ttest data=bark_beetle;

class treat;

var y;

run;

quit;

Figure 11.14: bark beetle experiment ttest.sas - proc print

Figure 11.15: bark beetle experiment ttest.sas - proc gplot

Figure 11.16: bark beetle experiment ttest.sas - proc ttest

11.7 Problems

1. A doctor conducts an experiment in which men are placed on four
different diets, consisting of a standard weight loss regimen (a control
treatment) and three new diets (Diets 1, 2, 3). The weight losses (lbs)
after six months are given in the following table.

Control Diet 1 Diet 2 Diet 3
19.5 20.0 20.8 25.9
20.5 16.4 17.4 25.9
16.6 11.9 16.7 25.8
19.3 22.1 16.8 22.5

(a) Test whether there is a significant difference among the four treat-
ments using one-way ANOVA, using manual calculations. Report
the P value and discuss the significance of the test, and then in-
terpret the results of the experiment. Show all your calculations.

(b) Repeat the analysis using SAS and proc glm. Attach your program
and output.

2. An experiment was conducted on the fecundity of a predatory insect
reared on an artificial diet using four different concentrations of the
preservative sorbic acid: (1) no sorbic acid, (2) 0.1% sorbic acid, (3)
0.2% sorbic acid, and (4) 0.5% sorbic acid. Twenty insects were reared
at each concentration and the fecundity of the resulting adults mea-
sured. See table below.

Treatment Observations
No sorbic acid 87, 124, 105, 87, 100, 89, 95, 79, 102, 112

92, 87, 115, 96, 111, 90, 86, 92, 109, 76
0.1% sorbic acid 105, 94, 97, 94, 83, 97, 107, 99, 104, 83

101, 71, 100, 75, 87, 106, 88, 99, 90, 74
0.2% sorbic acid 73, 94, 81, 83, 100, 98, 76, 91, 68, 82

92, 105, 76, 82, 95, 96, 101, 89, 92, 67
0.5% sorbic acide 83, 54, 86, 76, 74, 81, 79, 72, 80, 78

70, 83, 83, 85, 90, 70, 85, 94, 82, 75

Test whether there is a difference among the four treatments using one-
way ANOVA and SAS. Interpret the results of this analysis, providing
a P value and discussing the significance of the test. Using a graph,

explain what happens to fecundity as the concentration of sorbic acid
changes.

